PharmaSUG China 2019 - Paper DV-085

Visualization of efficacy endpoints in oncology clinical trials

Shilpakala Vasudevan, Ephicacy Lifescience Analytics

ABSTRACT

Visualization in the form of graphical representation is very important to understand and analyze clinical data. Especially for oncology studies, where survival of patients is critical, graphs greatly help in making decisions on the response of the treatments and their efficacy. Different kinds of plots are used for different kinds of efficacy endpoints, waterfall, spider, swimmer, kaplan meier, forest, spider plots.

INTRODUCTION

Oncology is one of the largest therapeutic areas in clinical trials. The nature of oncology data is quite complex and different as compared to the other therapeutic areas. Treatment response, tumor measurements and survival time of patients are the common analyses done in oncological studies. Graphical representation of these endpoints is an important part of reporting, for analyses and interpretation.

In this paper, we will see how we graphically represent some of the following widely used efficacy endpoints:

1. Best Overall Response
2. Objective Response Rate
3. Duration of Response
4. Overall Survival
5. Progression-Free Survival
6. Time to Progression

EFFICACY ENDPOINTS

BEST OVERALL RESPONSE

Best Overall Response is the best response recorded from the start of the study treatment until disease/progression. Best Overall Response could be Complete Response (CR), Partial Response (PR), Stable Disease (SD), or Progressive Disease (PD).

One of the ways to represent Response along with the change in size of tumor is with a waterfall plot.

Below dataset shows maximum percentage change from baseline and the best overall response for each subject:

Table 1: Dataset for waterfall plot for Best Overall Response

obs (Observation)	subjid (Subject Identifier)	maxchange (Maximum % Change from Baseline)	response (Best Overall Response)
1	192	92	Progressive Disease
2	243	80	Stable Disease
3	126	70	Progressive Disease
4	498	66	Progressive Disease
5	257	60	Progressive Disease
…		…	…
…		…	…

…		…	…
42	226	-88	Partial Response
43	743	-93	Progressive Disease
44	314	-95	Partial Response
45	118	-100	Complete Response

Above dataset is sorted in descending order, which is necessary before graphing a waterfall plot. The following procedure produces the waterfall plot.

```
PROC SGPLOT data=wfdata1;
  vbar n / response=change group=group name="response";
  refline 20 / lineattrs=(color=grey pattern=shortdash);
  refline -30 / lineattrs=(color=grey pattern=shortdash);
  keylegend "response" / location=inside position=bottom;
run;
```

In a waterfall plot, each subject is vertically represented as an individual bar. In the y axis, is the maximum percentage change from baseline in tumor measurement. The data is displayed in a descending order and gives the appearance of a waterfall.

A reference line is drawn at 0. For some subjects, maximum percent change from baseline may be above 0, which means there is an increase in tumor measurement. For others, change may be below 0, which indicates a decrease in tumor size. Along with change in tumor measurement, the best overall response is also shown by adding colors or patterns to the bars for each subject.

By looking at the below example waterfall plot, we can get a clear picture of the results from worst response to the best possible response of the subjects overall.

Figure 1: Waterfall plot for Best Overall Response



Another way of representing best overall response is through spider plot. Unlike a waterfall plot, where the overall maximum % change from baseline for every subject is displayed, in a spider plot we display % change from baseline for a subject at every significant timepoint.

Following is the dataset for plotting the spider plot.

Table 2: Dataset for spider plot for Best Overall Response

obs (Observation)	subjid (Subject ID)	month (Month)	Change (% Change from Baseline)	response (Best Overall Response)
1	101	0	0	Complete Response
2	101	5	-28	Complete Response
3	101	10	-10	Complete Response
4	101	15	-28	Complete Response
5	101	20	-100	Complete Response
6	101	25	-100	Complete Response
7	101	30	-100	Complete Response
8	102	0	0	Progressive Disease
9	102	5	30	Progressive Disease
10	102	10	14	Progressive Disease
11	102	15	38	Progressive Disease
12	102	20	90	Progressive Disease
13	102	0	0	Progressive Disease
14	102	5	30	Progressive Disease
15	102	10	14	Progressive Disease

Following is the code for spider plot:

```
PROC SGPLOT data=spdata;
  refline 0 / lineattrs=(color=grey pattern=solid);
  refline 20 / lineattrs=(color=grey pattern=dash);
  refline -30 / lineattrs=(color=grey pattern=dash);
  series x=month y=change / group=subjid grouplc=response groupmc= response
                            markers markerattrs=(symbol=circlefilled)
                            lineattrs=(thickness=2 pattern=solid);
run;
```

Here the change from baseline for each subject is plotted with time. Apart from 0, reference lines are also drawn at 20 and -30.

In spider plot, every subject is represented as a line that starts from the same point. Depending on whether there is an increase or decrease in tumor size, the lines then are plotted according to the change. This way the % change from baseline for each subject is plotted across the entire study duration. Here also, the subjects are color coded as per the best overall response recorded.

Spider plots are easier to interpret when there are not too many subjects. Otherwise the plot would look more cluttered and harder to read.

Figure 2: Spider plot for Best Overall Response



OBJECTIVE RESPONSE RATE

A patient is considered to have an objective response (OR) when the Best Overall Response of a subject is either a Partial Response (PR) or Complete Response (CR).

OR = (CR+PR)

The Objective Response Rate (ORR) is the percentage of patients with complete or partial response after treatment

ORR = (CR+PR) * 100/Total Number of Subjects

Following is the dataset, that has the mean and confidence limits for ORR:

Table 3: Dataset for bar graph for Objective Response Rate

obs (Observation)	trt (Treatment)	age (Age)	response (Response)	orr (Objective Response Rate)	orrmean (Mean)	lowerclm (lower CI)	upperclm (upper CI)
1	Treatment A	Age < 65 years	Partial Response	28	33.1	32.2	34.8
2	Treatment A	Age < 65 years	Complete Response	6	33.1	32.2	34.8
3	Treatment A	Age >= 65 years	Partial Response	23	25.5	22.9	27.8
4	Treatment A	Age >= 65 years	Complete Response	3	25.5	22.9	27.8
5	Treatment B	Age < 65 years	Partial Response	24	27.5	24.6	30.3

6	Treatment B	Age < 65 years	Complete Response	4	27.5	24.6	30.3
7	Treatment B	Age >= 65 years	Partial Response	17	19	18.3	22.4
8	Treatment B	Age >= 65 years	Complete Response	1	19	18.3	22.4

The above mean and confidence limits on the objective response rates can be calculated first by using proc means. Then proc sgplot is used to plot the data

```
PROC MEANS data=bardata1 mean lclm uclm;
  class trt;
  var orr;
  output out=bardata2 mean=orrmean lclm=lowerclm uclm=upperclm;
run;

PROC SGPLOT data=bardata3;
  panelby trt / layout=columnlattice colheaderpos=bottom;
  vbarparm category=trt response=orrmean / group=response limitlower=lowerclm
                                           limitupper=upperclm;
run;
```

Below graph displays objective response rate for both treatments for each age category. Each bar depicts the ORR for Partial Response and Complete Response, and for both together. Additionally, the 95% confidence intervals are also displayed for each bar. We can identify from the below example graph that Treatment A has higher objective response rates, indicating that there is better response from the treatment than Treatment B. Also, subjects of the age group <65 years show a better response compared to the patients in the other age category.

Figure 3: Bar Graph for Objective Response Rate



DURATION OF RESPONSE

Duration of Response (DOR) is the time from first documentation of response (partial or complete) to date of progression or death due to any cause.

DOR=Progression/death date – Date of first CR or PR + 1

Swimmer plots can be used for displaying duration of response for each subject. Below dataset has the start and end of every response for every subject:

Table 4: Dataset for swimmer plot for Duration of Response

subjid (Subject)	Low (low)	high (high)	highcap (highcap)	crstart1 (Start of Complete Response 1)	crstart2 (Start of Complete Response 2)	prstart1 (Start of Partial Response 1)	prstart2 (Start of Partial Response 2)	endresp1 (End of Response 1)	endresp2 (End of Response 2)
101	0	12		6.8	20.3			13.4	25
159	0	11.5	FilledArrow	8.9		26.2		17.7	
267	0	10.3				17.8		31.2	
295	0	9.5	FilledArrow			3.3	19.8	7.4	
412	0	9	FilledArrow	12.4					
434	0	8.2				15.7		24.5	
486	0	7.2	FilledArrow	1.8		17.3		6.2	
521	0	6.4	FilledArrow			9.1			
654	0	5.7	FilledArrow	7.8					
670	0	5				4.3		15.2	
738	0	4	FilledArrow			8.6			

```
PROC SGPLOT data=swdata1;
  highlow y=subjid low=low high=high / highcap=highcap type=bar;
  scatter y=subjid x=crstart1 / markerattrs=(symbol=trianglefilled
                                color=green);
  scatter y=subjid x=crstart2 / markerattrs=(symbol=trianglefilled
                                color=green);
  scatter y=subjid x=prstart1 / markerattrs=(symbol=trianglefilled
                                color=blue);
  scatter y=subjid x=prstart2 / markerattrs=(symbol=trianglefilled
                                color=blue);
  scatter y=subjid x=endresp1 / markerattrs=(symbol=circlefilled
                                color=black);
  scatter y=subjid x=endresp2 / markerattrs=(symbol= circlefilled
                                color=black);
run;
```

A swimmer plot has horizontal bars representing each subject that has experienced a response. Subjects that do not have a complete or partial response are not included in the plot.

For each subject, the start of a response and the end of a response is depicted inside the bar. The start of partial and complete responses is displayed with different kind of markers. For responses where the end is not known and continuing, it is displayed with a right-angled arrow in the bar. Swimmer plots help in graphically analyzing how long the responses last for the subjects.

Figure 4: Swimmer plot for Duration of Response



OVERALL SURVIVAL

Overall survival (OS) is the time from the start of treatment till the time of death.

OS = death date – start date + 1

Where death date is not known due to any reason, censoring is done and the last known alive date is taken instead of the death date. Overall Survival is considered to be the “gold standard” among primary efficacy endpoints.

Kaplan-Meier plot is the most popular tool for graphing survival analysis.

Table 5: Dataset for Kaplan Meier plot for Overall Survival

subjid (Subject Identifier)	trt (Treatment)	os (Overall Survival)	cnsr (Censoring done=1 Not done =0)
101	Treatment A	18	0
102	Treatment A	10	1
103	Treatment B	11	1
104	Treatment A	4	0
105	Treatment B	31	0
106	Treatment B	9	1

The first step is to use proc lifetest to generate the Kaplan Meier estimates.

```
ods listing close;
ods output productlimitestimates=kmdata2;
PROC LIFETEST data=kmdata1 method=km outsurv=kmci;
  time os*cnsr(1);
  by trtn trt;
run;
ods output close;
ods listing;
```

Additionally, cox regression analysis can be done to derive hazard ratio, and log rank tests to generate p values.

```
PROC PHREG data=kmdata1;
  class trt(ref='Treatment B');
  model os*cnsr(1) = trt / risklimits;
  hazardratio trt;
run;

PROC SGPLOT data=kmdata3;
  step x=os y=survival / group=trt;
  scatter x=os y=censored / markerattrs=(symbol=circle) group=trt;
run;
```

The below example graph shows the survival rates of subjects in both treatments at every time point. Also, subjects that did not experience an event and are censored are displayed with markers in the plot. From the below graph, we can deduce that Treatment A has better overall survival rate than the other.

Figure 5: Kaplan-Meier plot for Overall Survival



PROGRESSION-FREE SURVIVAL

Progression-free survival is the other gold standard endpoint in oncology. Progression-Free Survival (PFS) is the time from the start of treatment to the first documentation of progression or death.

PFS = progression or death date – start date + 1

Where there is no progression or death date, censoring is done, and dates are taken as per the censoring rules of the study.

PFS endpoints can be graphically represented with Kaplan-Meier plots. In addition, forest plots are used to represent this endpoint. Following is the dataset that has the hazard ratio and confidence intervals, by subgroup:

Table 6: Dataset for Forest plot for Progression-Free Survival

obs (Observation)	subgroup (Subgroup)	N1 (Treatment A Count)	N2 (Treatment B Count)	hratio (Hazard Ratio)	cilow (lower CI)	cihigh (higher CI)
1	Overall	100	100	0.7	0.61	1.05
2	Age					
3	<65	45	50	0.8	0.55	1.13
4	>=65	55	50	0.89	0.57	1.3
5	Gender					
6	Male	62	65	0.52	0.3	1.11
7	Female	38	35	1.08	0.82	1.5
8	ECOG Performance Status					
9	0	57	55	0.75	0.63	0.9
10	1	43	45	1.25	1.3	1.62
11	Measurable disease					
12	Measurable	79	72	0.8	0.45	1.2
13	Non-measurable	21	28	1.45	1.15	1.75

```
PROC SGPLOT data=frstdata1;
  highlow y=obs low=cilow high=cihigh;
  scatter y=obs x=hratio / markerattrs=(symbol=squarefilled);
  scatter y=obsid x=hratio / markerattrs=(size=0) x2axis;
  refline 1 / axis=x;
  yaxistable subgroup / location=inside position=left;
  yaxistable N1 / location=inside position=left;
  yaxistable N2 / location=inside position=left;

  xaxis display=(nolabel) values=(0.0 0.5 1.0 1.5 2.0);
  x2axis display=(noline noticks novalues);
run;
```

Forest plots are useful if we would like to do subgroup analyses. It's also very useful when we would like to view the results as a table and a graph together in one output.

For every subgroup in the below graph, hazard ratio and confidence intervals are displayed in each line. The below example graph indicates that the results favor Treatment A over Treatment B.

Figure 6: Forest plot for Progression-Free Survival



TIME TO PROGRESSION

Time To Progression (TTP) is the time from start date to the date of first documentation of objective progression.

TTP = progression date – start date + 1

Table 7: Dataset for Skyline plot for Time To Progression

obs (Observation)	months (Study Duration in months)	hratio (Hazard Ratio)	cilow (lower CI)	cihigh (higher CI)
1	1	0.75	0.38	1.32
2	3	0.6	0.29	0.85
3	5	0.52	0.32	0.82
4	8	0.73	0.42	1.0
5	10	0.78	0.5	1.1
6	12	0.65	0.48	1.12
7	13	0.77	0.47	1.1
8	15	0.65	0.43	1.15
9	17	0.75	0.53	1.2
10	20	0.72	0.48	0.93

```
PROC SGPLOT data=skyldata;
  band x=months lower=cilow upper=cihigh;
```

```
  series x=months y=hratio / lineattrs=(thickness=2px);
  refline 1 / axis=y;
run;
```

In below skyline plot, time to progression is plotted over time. The hazard ratio is also depicted along with the confidence intervals.

Figure 7: Skyline plot for Time To Progression



CONCLUSION

This paper has presented some widely used means of representing efficacy endpoints in oncology. The representation of these endpoints is not restricted to the way it is shown in this paper. There may be multiple ways of depicting endpoints in a study. For example, progression free survival can be either displayed using a Kaplan-Meier plot or a Forest plot or a Skyline plot, depending on the need of the analyses.

Besides the endpoints reviewed in this paper, there are other efficacy endpoints too in oncology. These too could be displayed using similar tools as discussed in this paper. For example, Best Clinical Benefit Response (BCBR) could be represented in the same way as Best Overall Response (BOR); Time to Failure (TTF) could be reported in the same kind as Time to Progression (TTP); and Event-Free Survival (EFS) and Disease-Free Survival (DFS) in a similar approach as Progression-Free Survival (PFS).

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Shilpakala Vasudevan
Ephicacy Lifescience Analytics
shilpakala.vasudevan@ephicacy.in